
Appel d'un écran de sélection 
 
 
CMD : Exemple               
              CMD        PROMPT(CL)                                         
              PARM       KWD(SOC) TYPE(*CHAR) LEN(2) CONSTANT('  ')           
              PARM       KWD(ETAB) TYPE(*CHAR) LEN(2) CONSTANT('  ')             
              PARM       KWD(DPERI) TYPE(*CHAR) LEN(6) CONSTANT('    ')           
              PARM       KWD(FPERI) TYPE(*CHAR) LEN(6) CONSTANT('    ')           
 
CL : Exemple               
Quand le CL est appelé sous le mode interactif les paramètres sont entrés à blanc.   
Quand le CL est appelé sous le mode batch les paramètres sont  chargés.  
 
   1600              PGM        PARM(&BSOC &BETAB &BPERI &BFPERI)               
   1700                                                                        
   1900 /*  Déclaration de l'écran  */  
   2200              DCLF       FILE(DSPF) 
   2300                                         
   2500 /*  Déclaration des variables  */  
   2800              DCL        VAR(&ERR)     TYPE(*CHAR) LEN(50)   
   2900                                                             
   3000              DCL        VAR(&£ETAB)   TYPE(*CHAR) LEN(2)  
   3100              DCL        VAR(&YM)      TYPE(*CHAR) LEN(10)       /* WRKDATE    */  
   3200              DCL        VAR(&YMD)     TYPE(*CHAR) LEN(10)                        
   3300              DCL        VAR(&DMY)     TYPE(*CHAR) LEN(10)                      
   3400              DCL        VAR(&ERROR)   TYPE(*CHAR) LEN(57)                   
   3500              DCL        VAR(&OPERI)   TYPE(*CHAR) LEN(16)                   
   3600              DCL        VAR(&OPERD)   TYPE(*CHAR) LEN(16)                   
   3700              DCL        VAR(&DATES)   TYPE(*CHAR) LEN(8)                   
   3800              DCL        VAR(&DPERD)   TYPE(*CHAR) LEN(6)                  
   3900              DCL        VAR(&FPERF)   TYPE(*CHAR) LEN(6)                  
   4000                                                                    
   4100              DCL        VAR(&PERIOD1) TYPE(*CHAR) LEN(17)           
   4200              DCL        VAR(&PERIOD2) TYPE(*CHAR) LEN(17)          
   4300                                                                                     
   4400              DCL        VAR(&ANNEE)   TYPE(*CHAR) LEN(2)        /* RTVSYSVAL  */   
   4500              DCL        VAR(&MOIS)    TYPE(*CHAR) LEN(2)                        
   4600              DCL        VAR(&AAMM)    TYPE(*CHAR) LEN(4)                         
   4700              DCL        VAR(&AA)      TYPE(*DEC)  LEN(2 0)                       
   4800                                                                                   
   4900              DCL        VAR(&TOUCH)   TYPE(*CHAR) LEN(1)        /* ANRCFR10   */  
   5000                                                                                   
   5100              DCL        VAR(&TYPE)   TYPE(*CHAR) LEN(1)         /* RTVJOBA    */    
   5200                                                                                     
   5300              DCL        VAR(&PRINT)  TYPE(*CHAR) LEN(10)        /* CHOIX IMP. */    
   5400              DCL        VAR(&DTAARA) TYPE(*CHAR) LEN(10)                          
   5500              DCL        VAR(&BIB)   TYPE(*CHAR) LEN(10)                            
   5600 /* Standard Gestion des erreurs */                         
   5700  DCLMSG:     DCL  &MSG     *CHAR  80                                         
   5800              DCL  &MSGF    *CHAR  10                                      
   5900              DCL  &MSGFLIB *CHAR  10                                      
   6000              DCL  &MSGID   *CHAR   7                                       
   6100              DCL  &MSGDTA  *CHAR  80                               
   6200                                                                         
   6300              DCL  &RTNTYPE *CHAR   2          /* Type de retour message   */   
   6400              DCL  &DIAG    *CHAR   2  '02'                                   
   6500              DCL  &EXCP    *CHAR   2  '15'                                   
   6600              DCL  &CANCELD *CHAR  30  'Commande annulée'                         
   6700                                                                                       
   6900 /* Moniteur général d'erreurs et routine de gestion des erreurs  */        
   7200              MONMSG     MSGID(CPF0000 MCH0000) EXEC(GOTO +                       
   7400                                                                               
   7500 /* ------------------------------------------------------------------------ */  
   7600 /* Extraire attribut, valeur */       
   7900     /* recherche si job est interactif ou batch */  
   8100              RTVJOBA    TYPE(&TYPE)                            
   8200                                                               
   8300     /* charger des valeurs par défaut à l'écran */                
   8500              IF         COND(&TYPE *EQ '1') THEN(DO)            /* Début INTERACTIF */   

   8600                                                                                  
   8700              CHGVAR     &BSOC  %SST(*LDA 1 2)                    
   8800              CHGVAR     &BETAB %SST(*LDA 3 2)                   
   8900              CHGVAR     &APPL %SST(*LDA 31 3)                    
   9000              CHGVAR     &NDOS %SST(*LDA 90 4)                       
   9100                                                                                   
   9200              RTVSYSVAL  SYSVAL(QYEAR) RTNVAR(&ANNEE)            /* période    */   
   9300              RTVSYSVAL  SYSVAL(QMONTH) RTNVAR(&MOIS)                             
   9400              CHGVAR     VAR(&AAMM) VALUE(&ANNEE *CAT &MOIS)                      
   9500              CHGVAR     VAR(&OPERD) VALUE(&ANNEE *CAT &MOIS *CAT '        -' +   
   9600                         *CAT '001')                                            
   9700              WRKDATE    FROMFMT($YM) DATE(&OPERD) TOFMT($YMC) +               
   9800                           TODATE(&OPERI) ERREUR(&ERROR)                     
   9900              CHGVAR     VAR(&BPERI) VALUE(%SST(&OPERI 1 6))                 
  10000              CHGVAR     VAR(&BFPERI) VALUE(%SST(&OPERI 1 6))               
  10100                                                                     
  10200     /* recherche de la dtaara et création si inexistante */         
  10300              RTVDTAARA  DTAARA(APAGRP00 (1 10)) RTNVAR(&BIB)          
  10400              CHGVAR     VAR(&DTAARA) VALUE('ANACFR10')                  
  10500              CHKOBJ     OBJ(&BIB/&DTAARA) OBJTYPE(*DTAARA)              
  10600              MONMSG     MSGID(CPF9801) EXEC(CRTDTAARA +                 
  10700                           DTAARA(&BIB/&DTAARA) TYPE(*CHAR) LEN(10) +    
  10800                           TEXT('Libellé de la dtaara') AUT(*ALL))       
  11000                                                                
  11100              RTVDTAARA  DTAARA(&BIB/&DTAARA) RTNVAR(&PRINT)    /* imprimante */ 
  11200     /* si aucune imprimante, QPRINT par défaut */                                                          
  11300              IF         COND(&PRINT = ' ')  THEN(DO)        
  11400              CHGVAR     VAR(&PRINT)   VALUE(QPRINT)        
  11500              ENDDO                                         
  11600                                                                /* Fin INTERACTIF */      
  11700              ENDDO                                                                  
  11800                                                                                    
  12000 /*  Traitement de l'écran  */  
  12200                                                                                       
  12300              IF         COND(&TYPE *EQ '1') THEN(DO)       /* TYPE INTERACTIF */     
  12400  £PROMPT:                                                                         
  12500              SNDRCVF    DEV(*FILE) RCDFMT(ECRAN)                                  
  12600                                                                                  
  12700              RMVMSG     CLEAR(*ALL)                                              
  12800              CHGVAR     VAR(&IN99)   VALUE('0')                 /* indic. écr */  
  12900              CHGVAR     VAR(&IN60)   VALUE('0')                                   
  13000              CHGVAR     VAR(&IN61)   VALUE('0')                                   
  13100              CHGVAR     VAR(&IN62)   VALUE('0')                                   
  13200              CHGVAR     VAR(&IN63)   VALUE('0')                                   
  13300              CHGVAR     VAR(&IN64)   VALUE('0')                                   
  13400              CHGVAR     VAR(&ERRMSG) VALUE(' ')                 /* zone erreu */  
  13500                                                                                  
  13600     /* si F03 = 1 fin de programme */                                            
  13700                                                                                 
  13800              IF         COND(&IN03 *EQ '1' *OR &IN12 *EQ '1' ) THEN(DO)      
  13900              SNDPGMMSG  MSG('** Demande Annulée ou Terminée')               
  14000              GOTO       £ENDPGM                                            
  14100              ENDDO                                                         
  14200                                                                         
  14300     /* test */                                           
  14500              CHGVAR     VAR(&ERR) VALUE('RSC')                                      
  14600              CHGVAR VAR(&£ETAB) VALUE(&BETAB)                                  
  14700              IF COND(&BETAB *EQ '  ') THEN(CHGVAR VAR(&£ETAB) VALUE(&BSOC))      
  14800              CALL       PGM(TEST1) PARM(&BSOC &£ETAB 'ANA ' &ERR)           
  14900                                                                                      
  15000              IF         COND(&ERR *EQ '*ERREUR') THEN(DO)       /* ERREUR     */     
  15100              CHGVAR     VAR(&IN99) VALUE('1')                                     
  15200              CHGVAR     VAR(&IN60) VALUE('1')                                      
  15300              CHGVAR     VAR(&IN61) VALUE('1')                                     
  15400              CHGVAR     VAR(&ERRMSG) VALUE('Donnée incorrecte')       
  15600              GOTO       £PROMPT                                            
  15700              ENDDO                                                           
  15800                                                                                
  15900     /* test de la période */                                                    
  16300              WRKDATE    FROMFMT($YMC) DATE(&BPERI) TOFMT($YMC) +                       
  16400                           TODATE(&OPERI) ERREUR(&ERROR)              /* début */        

  16500                                                                                        
  16600              IF COND(&ERROR *NE ' ') THEN(DO)                   /* erreur     */        
  16700              CHGVAR     VAR(&IN99) VALUE('1')                                       
  16800              CHGVAR     VAR(&IN62) VALUE('1')                                       
  16900              CHGVAR     VAR(&ERRMSG) VALUE('La période est incorrecte')             
  17000              GOTO       £PROMPT                                             
  17100              ENDDO                                                            
  17200              CHGVAR     VAR(&DPERD) VALUE(%SST(&OPERI 1 6))            
  17300                                                                            
  17400              WRKDATE    FROMFMT($YMC) DATE(&BFPERI) TOFMT($YMC) +         
  17500                           TODATE(&OPERI) ERREUR(&ERROR)              /* fin   */  
  17600                                                                                    
  17700              IF COND(&ERROR *NE ' ') THEN(DO)                   /* erreur     */  
  17800              CHGVAR     VAR(&IN99) VALUE('1')                                     
  17900              CHGVAR     VAR(&IN63) VALUE('1')                                     
  18000              CHGVAR     VAR(&ERRMSG) VALUE('La période est incorrecte')          
  18100              GOTO       £PROMPT                                                 
  18200              ENDDO                                                   
  18300              CHGVAR     VAR(&FPERF) VALUE(%SST(&OPERI 1 6))      
  18400                                                                 
  18500              IF         COND(&DPERD *GT &FPERF) THEN(DO)        
  18600              CHGVAR     VAR(&IN99) VALUE('1')                      
  18700              CHGVAR     VAR(&IN63) VALUE('1')                       
  18800              CHGVAR     VAR(&ERRMSG) VALUE('La période de fin est +    
  18900                           inférieure à la période début')          
  19000              GOTO       £PROMPT                                    
  19100              ENDDO                                         
  19200                                                              
  19300     /* test de l'imprimante  */                                                     
  19500              IF         COND(&PRINT = ' ')  THEN(DO)      
  19600              CHGVAR     VAR(&PRINT)   VALUE(QPRINT)         
  19700              ENDDO                                         
  19800                                                                   
  19900              IF   COND(&PRINT *NE 'QPRINT') THEN(DO)            
  20000              CHKOBJ     OBJ(&PRINT) OBJTYPE(*OUTQ) AUT(*USE)   
  20100              MONMSG     MSGID(CPF0000) EXEC(DO)                 
  20200              CHGVAR     VAR(&IN99) VALUE('1')                    
  20300              CHGVAR     VAR(&IN64)   VALUE('1')                  
  20400              RCVMSG     MSGTYPE(*ANY) RMV(*YES) MSG(&ERRMSG)      
  20500              GOTO £PROMPT                                          
  20600              ENDDO                                                  
  20700              ENDDO                                                      
  20800                                                                       
  20900     /* mise en mémoire de l'imprimante dans la dtaara */           
  21100              CHGDTAARA  DTAARA(&BIB/&DTAARA) VALUE(&PRINT)              
  21200                                                                                        
  21300              ENDDO                                    /* FIN TYPE  INTERACTIF */     
  21400                                                                                       
  21600 /*  Auto-soumission du pgm pour édition */                                         
  21900              IF         COND(&BEDIT = 'O') THEN(DO)                     
  22000              SBMJOB     CMD(CALL PGM(EDITION) PARM(&BSOC &BETAB +      
  22100                           &BPERI &BFPERI)) JOB(EDIT) OUTQ(&PRINT) +    
  22200                           LOG(*JOBD *JOBD *NOLIST)                     
  22300              CHGVAR     VAR(&ERRMSG) VALUE('Edition envoyée sur +        
  22400                           l''imprimante ' *CAT &PRINT *CAT ' ')          
  22500              GOTO £PROMPT                                                      
  22600              ENDDO                                                     
  22700                                                                       
  22900 /*  Préparation des variables pour OPNQRYF  */     
  23200     /* date début à partir de la période BPERI */        
  23300              CHGVAR     &DATES (&BPERI *CAT '01')        
  23400              WRKDATE    FROMFMT(*YMDC) DATE(&DATES) TOFMT(*YMDC) +   
  23500                           TODATE(&OPERI) ERREUR(&ERROR)              
  23600              WRKDATE    FROMFMT(£LIBP) DATE(&OPERI) TOFMT(*YMDC) +     
  23700                           TODATE(&OPERI) ERREUR(&ERROR)               
  23800              CHGVAR     VAR(&PERIOD1) VALUE(%SST(&ERROR 1 17))   /* Libellé date */ 
  23900                                                                                      
  24000     /* date fin à partir de la période   BFPERI */                
  24100              CHGVAR     &DATES (&BFPERI *CAT '01')                  
  24200              WRKDATE    FROMFMT(*YMDC) DATE(&DATES) TOFMT(*YMDC) +   
  24300                           TODATE(&OPERI) ERREUR(&ERROR)               

  24400              WRKDATE    FROMFMT(£DJDM) DATE(&OPERI) TOFMT(*YMDC) +   
  24500                           TODATE(&OPERI) ERREUR(&ERROR)              
  24600              WRKDATE    FROMFMT(£LIBP) DATE(&OPERI) TOFMT(*YMDC) +   
  24700                           TODATE(&OPERI) ERREUR(&ERROR)               
  24800              CHGVAR     VAR(&PERIOD2) VALUE(%SST(&ERROR 1 17))   /* Libellé date */     
  24900                                                                                     
  25000              CHGVAR     &DATES (&BPERI *CAT '01')                             
  25100                                                                            
  25200              SNDPGMMSG  MSGID(CPF9898) MSGF(QCPFMSG) MSGDTA('Demande +    
  25300                           en cours de traitement...') TOPGMQ(*EXT) +    
  25400                           MSGTYPE(*STATUS)                             
  25500                                                                      
  25600              CHGJOB     STSMSG(*NONE)                                  
  25700                                                                                 
  25800     /* sélection  */     
  25900              WRKDATE    FROMFMT($YMC) DATE(&BPERI) TOFMT($YMC) +                    
  26000                           TODATE(&OPERI) ERREUR(&ERROR)              /* début */    
  26100              CHGVAR     VAR(&DPERD) VALUE(%SST(&OPERI 1 6))                          
  26200              WRKDATE    FROMFMT($YMC) DATE(&BFPERI) TOFMT($YMC) +                  
  26300                           TODATE(&OPERI) ERREUR(&ERROR)              /* fin   */  
  26400              CHGVAR     VAR(&FPERF) VALUE(%SST(&OPERI 1 6))                       
  26500                                                                           
  26600              OVRDBF     FILE(Fichier) SHARE(*YES)              
  26700                                                                      
  26800              IF COND(&BETAB *NE '  ') THEN( +                         
  26900              OPNQRYF    FILE((Fichier)) QRYSLT('SOC *EQ "' *CAT +    
  27000                           &BSOC *CAT '" *AND ETAB *EQ "' *CAT +       
  27100                           &BETAB *CAT '" *AND %DIGITS(MCUM) *GE "' +   
  27200                           *CAT &DPERD *CAT '" *AND %DIGITS(MCUM) +     
  27300                           *LE "' *CAT &FPERF *CAT '"') +               
  27400                           KEYFLD((K1) (K2) (K3)))                
  27500                                                                            
  27600              IF         COND(&BETAB *EQ '  ') THEN(OPNQRYF +               
  27700                           FILE((Fichier)) QRYSLT('SOC *EQ "' *CAT +     
  27800                           &BSOC *CAT '" *AND %DIGITS(MCUM) *GE "' +       
  27900                           *CAT &DPERD *CAT '" *AND %DIGITS(MCUM) +       
  28000                           *LE "' *CAT &FPERF *CAT '"') +                   
  28100                           KEYFLD((K1) (K2) (K3)))                         
  28200                                                                                    
  28400              CHGVAR     VAR(&TOUCH) VALUE(' ')                             
  28500                                                                            
  28600     /* appel du pgm pour afficher ou éditer le cumul */                     
  28800              IF         COND(&TYPE *EQ '1') THEN(DO)                       
  28900              CALL       PGM(Affichage) PARM(&BSOC &BETAB &PERIOD1 +        
  29000                           &PERIOD2 &TOUCH)                        /* afficher */    
  29100              IF         COND(&TOUCH *EQ '2') THEN(DO)                       
  29200              CHGVAR     VAR(&TOUCH) VALUE('1')                              
  29300              CHGVAR     VAR(&IN99) VALUE('1')                       
  29400              CHGVAR     VAR(&ERRMSG) VALUE('Aucun enregistrement +     
  29500                           pour la période demandée.')                 
  29600              ENDDO                                                     
  29700              ENDDO                                                     
  29800              ELSE (DO)                                                    
  29900              CALL       PGM(Edition) PARM(&BSOC &BETAB &PERIOD1 &PERIOD2)  /* éditer   
*/   
  30100              ENDDO                                                                  
  30200                                                                                      
  30300              CHGJOB     STSMSG(*USRPRF)                                          
  30400              CLOF         OPNID(Fichier)                                      
  30500              DLTOVR       FILE(*ALL)           
  30600                                                    
  30700     /* si F12 (venant de Affichage) */                    
  30900              IF         COND(&TOUCH *EQ '1') THEN(DO)       
  31000              RCLRSC                                      
  31100              GOTO       £PROMPT                           
  31200              ENDDO                                        
  31300                                                    
  31500 /* TRAITEMENT DES ERREURS --------------------------------------------------------- */  
  31800  ERREUR:                                                                                  
  32000     /* Réception des Messages */                                                          
  32200  RCVMSG:     RCVMSG     MSGDTA(&MSGDTA) MSGID(&MSGID) +                                   

  32300                           RTNTYPE(&RTNTYPE) MSGF(&MSGF) +                                 
  32400                           MSGFLIB(&MSGFLIB)                                               
  32600     /* Renvoi Messages */                                                                 
  32700                                                                                           
  32800              IF         COND(&MSGID *NE ' ') THEN(DO)                                     
  32900              IF         COND((&RTNTYPE = &DIAG) *OR (&RTNTYPE = +                         
  33000                           &EXCP)) THEN(DO)                                                
  33100              SNDPGMMSG  MSGID(&MSGID) MSGF(&MSGFLIB/&MSGF) +                              
  33200                           MSGDTA(&MSGDTA) TOPGMQ(*PRV) MSGTYPE(*DIAG)                     
  33300              MONMSG     MSGID(CPF0000)                                                    
  33400              ENDDO                                                                        
  33500              GOTO       CMDLBL(RCVMSG)                                                    
  33600              ENDDO                                                                        
  33700              ELSE       CMD(DO)                                                           
  33900     /* Message de Fin anormale */                                                         
  34000                                                                                           
  34100              SNDPGMMSG  MSGID(CPF9898) MSGF(QCPFMSG) +                                        
  34200                           MSGDTA(&CANCELD) TOPGMQ(*PRV) MSGTYPE(*DIAG)                        
  34300              MONMSG     MSGID(CPF0000)                                                        
  34400              ENDDO                                                                            
        /* -------------------------------------------------------------------------------- */ 
  34600  £ENDPGM:    RCLRSC                                                                           
  34700              ENDPGM                                                                           
 
DSPF : Exemple               
     30      A                                      DSPSIZ(24 80 *DS3)    
     40      A                                      CHGINPDFT(CS UL)       
     50      A          R ECRAN                                             
     70      A                                      CF03(03 'Fin')           
     80      A                                      CF12(12)                 
     90      A                                      BLINK                     
    100      A                                      PRINT                     
    110      A                                  1 64DATE                      
    120      A                                      EDTCDE(Y)                 
    130      A                                  1 73TIME                       
    140      A                                 22 10'F3=Fin'                   
    150      A                                      COLOR(BLU)                 
    160      A            ERRMSG        70A  O 24  4                            
    170      A  99                                  DSPATR(RI)                 
    180      A                                      COLOR(RED)                
    190      A                                  9 25'Sté / Ets  . . . .'        
    200      A                                 11 25'Période  . . .  du'        
    210      A            BSOC           2A  B  9 44                            
    220      A  60                                  DSPATR(RI)                  
    230      A  60                                  DSPATR(PC)                  
    240      A                                      COLOR(BLU)                  
    250      A            BETAB          2A  B  9 51                            
    260      A  61                                  DSPATR(RI)                   
    270      A  61                                  DSPATR(PC)                   
    280      A                                      COLOR(BLU)                   
    290      A            BPERI          6A  B 11 44                             
    300      A  62                                  DSPATR(RI)                   
    310      A  62                                  DSPATR(PC)                   
    320      A                                      COLOR(YLW)                    
    330      A                                  1  2'CFR10'                     
    340      A                                 14 25'Edition  . . . . .'       
    350      A            BEDIT          1A  B 14 45COLOR(PNK)                 
    360      A                                      VALUES('N' 'O')            
    370      A                                      DFTVAL('N')                
    380      A                                 14 60'(O/N)'                    
    390      A                                      COLOR(BLU)                 
    400      A                                 11 52'au'                      
    410      A            BFPERI         6A  B 11 56                          
    420      A  63                                  DSPATR(RI)                    
    430      A  63                                  DSPATR(PC)                    
    440      A                                      COLOR(YLW)                         
    450      A            PRINT         10A  B 16 45                                      
    460      A  64                                  DSPATR(RI)                            
    470      A  64                                  DSPATR(PC)                          
    500      A                                 22 29'F12=Retour'                    
    510      A                                      COLOR(BLU)                     

    520      A                                  1 30' Titre '       
    530      A                                      DSPATR(HI)                      
    540      A                                      DSPATR(UL)                   
    550      A                                  3 36'Cumuls de Périodes'            
    560      A                                      DSPATR(UL)                         
    570      A                                  9 48'/'                                   
    580      A                                 11 65'(aaaamm)'                       
    590      A                                      COLOR(BLU)       
 
 


